NAG Library Routine Document

C05PCF/C05PCA

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

C05PCF/C05PCA is a comprehensive routine that finds a solution of a system of nonlinear equations by a modification of the Powell hybrid method. You must provide the Jacobian.

2 Specification

2.1 Specification for C05PCF

```
SUBROUTINE C05PCF (FCN, N, X, FVEC, FJAC, LDFJAC, XTOL, MAXFEV, DIAG, MODE,     &
                   FACTOR, NPRINT, NFEV, NJEV, R, LR, QTF, W, IFAIL)

INTEGER            N, LDFJAC, MAXFEV, MODE, NPRINT, NFEV, NJEV, LR, IFAIL
REAL (KIND=nag_wp) X(N), FVEC(N), FJAC(LDFJAC,N), XTOL, DIAG(N), FACTOR,       &
                   R(N*(N+1)/2), QTF(N), W(1,1)
EXTERNAL           FCN
```

2.2 Specification for C05PCA

```
SUBROUTINE C05PCA (FCN, N, X, FVEC, FJAC, LDFJAC, XTOL, MAXFEV, DIAG, MODE,     &
                   FACTOR, NPRINT, NFEV, NJEV, R, LR, QTF, W, IUSER, RUSER,     &
                   IFAIL)

INTEGER            N, LDFJAC, MAXFEV, MODE, NPRINT, NFEV, NJEV, LR,             &
                   IUSER(*), IFAIL
REAL (KIND=nag_wp) X(N), FVEC(N), FJAC(LDFJAC,N), XTOL, DIAG(N), FACTOR,       &
                   R(N*(N+1)/2), QTF(N), W(1,1), RUSER(*)
EXTERNAL           FCN
```

3 Description

The system of equations is defined as:

$$f_i(x_1, x_2, \ldots, x_n) = 0, \qquad i = 1, 2, \ldots, n.$$

C05PCF/C05PCA is based on the MINPACK routine HYBRJ (see Moré *et al.* (1980)). It chooses the correction at each step as a convex combination of the Newton and scaled gradient directions. The Jacobian is updated by the rank-1 method of Broyden. At the starting point, the Jacobian is calculated, but it is not recalculated until the rank-1 method fails to produce satisfactory progress. For more details see Powell (1970).

4 References

Moré J J, Garbow B S and Hillstrom K E (1980) User guide for MINPACK-1 *Technical Report ANL-80-74* Argonne National Laboratory

Powell M J D (1970) A hybrid method for nonlinear algebraic equations *Numerical Methods for Nonlinear Algebraic Equations* (ed P Rabinowitz) Gordon and Breach

5 Parameters

1: FCN – SUBROUTINE, supplied by the user. *External Procedure*

Depending upon the value of IFLAG, FCN must either return the values of the functions f_i at a point x or return the Jacobian at x.

The specification of FCN for C05PCF is:

```
SUBROUTINE FCN (N, X, FVEC, FJAC, LDFJAC, IFLAG)

INTEGER            N, LDFJAC, IFLAG
REAL (KIND=nag_wp) X(N), FVEC(N), FJAC(LDFJAC,N)
```

The specification of FCN for C05PCA is:

```
SUBROUTINE FCN (N, X, FVEC, FJAC, LDFJAC, IFLAG, IUSER, RUSER)

INTEGER            N, LDFJAC, IFLAG, IUSER(*)
REAL (KIND=nag_wp) X(N), FVEC(N), FJAC(LDFJAC,N), RUSER(*)
```

1: N – INTEGER *Input*

On entry: n, the number of equations.

2: X(N) – REAL (KIND=nag_wp) array *Input*

On entry: the components of the point x at which the functions or the Jacobian must be evaluated.

3: FVEC(N) – REAL (KIND=nag_wp) array *Input/Output*

On entry: if IFLAG $= 0$ or 2, FVEC contains the function values $f_i(x)$ and must not be changed.

On exit: if IFLAG $= 1$ on entry, FVEC must contain the function values $f_i(x)$ (unless IFLAG is set to a negative value by FCN).

4: FJAC(LDFJAC,N) – REAL (KIND=nag_wp) array *Input/Output*

On entry: if IFLAG $= 0$, FJAC(i,j) contains the value of $\dfrac{\partial f_i}{\partial x_j}$ at the point x, for $i = 1,2,\ldots,n$ and $j = 1,2,\ldots,n$. When IFLAG $= 0$ or 1, FJAC must not be changed.

On exit: if IFLAG $= 2$ on entry, FJAC(i,j) must contain the value of $\dfrac{\partial f_i}{\partial x_j}$ at the point x, for $i = 1,2,\ldots,n$ and $j = 1,2,\ldots,n$, (unless IFLAG is set to a negative value by FCN).

5: LDFJAC – INTEGER *Input*

On entry: the first dimension of the array FJAC as declared in the (sub)program from which C05PCF/C05PCA is called.

6: IFLAG – INTEGER *Input/Output*

On entry: IFLAG $= 0$, 1 or 2.

IFLAG $= 0$

X and FVEC are available for printing (see NPRINT).

IFLAG $= 1$

FVEC is to be updated.

IFLAG $= 2$

FJAC is to be updated.

On exit: in general, IFLAG should not be reset by FCN. If, however, you wish to terminate execution (perhaps because some illegal point X has been reached), then IFLAG should be set to a negative integer. This value will be returned through IFAIL.

Note: *the following are additional parameters for specific use with C05PCA. Users of C05PCF therefore need not read the remainder of this description*.

7: IUSER(∗) – INTEGER array *User Workspace*
8: RUSER(∗) – REAL (KIND=nag_wp) array *User Workspace*

FCN is called with the parameters IUSER and RUSER as supplied to C05PCF/C05PCA. You are free to use the arrays IUSER and RUSER to supply information to FCN as an alternative to using COMMON global variables.

On entry: IFLAG = 0, 1 or 2.

IFLAG = 0
X and FVEC are available for printing (see NPRINT).

IFLAG = 1
FVEC is to be updated.

IFLAG = 2
FJAC is to be updated.

On exit: in general, IFLAG should not be reset by FCN. If, however, you wish to terminate execution (perhaps because some illegal point X has been reached), then IFLAG should be set to a negative integer. This value will be returned through IFAIL.

FCN must either be a module subprogram USEd by, or declared as EXTERNAL in, the (sub)program from which C05PCF/C05PCA is called. Parameters denoted as *Input* must **not** be changed by this procedure.

2: N – INTEGER *Input*

On initial entry: n, the number of equations.

Constraint: N $>$ 0.

3: X(N) – REAL (KIND=nag_wp) array *Input/Output*

On entry: an initial guess at the solution vector.

On exit: the final estimate of the solution vector.

4: FVEC(N) – REAL (KIND=nag_wp) array *Output*

On exit: the function values at the final point returned in X.

5: FJAC(LDFJAC,N) – REAL (KIND=nag_wp) array *Output*

On exit: the orthogonal matrix Q produced by the QR factorisation of the final approximate Jacobian.

6: LDFJAC – INTEGER *Input*

On entry: the first dimension of the array FJAC as declared in the (sub)program from which C05PCF/C05PCA is called.

Constraint: LDFJAC $\geq$ N.

7: XTOL – REAL (KIND=nag_wp) *Input*

On entry: the accuracy in X to which the solution is required.

Suggested value: $\sqrt{\epsilon}$, where ϵ is the ***machine precision*** returned by X02AJF.

Constraint: XTOL ≥ 0.0.

8: MAXFEV – INTEGER *Input*

On entry: the maximum number of calls to FCN with IFLAG $\neq 0$. C05PCF/C05PCA will exit with IFAIL $= 2$, if, at the end of an iteration, the number of calls to FCN exceeds MAXFEV.

Suggested value: MAXFEV $= 100 \times (\mathrm{N} + 1)$.

Constraint: MAXFEV > 0.

9: DIAG(N) – REAL (KIND=nag_wp) array *Input/Output*

On entry: if MODE $= 2$, DIAG must contain multiplicative scale factors for the variables.

If MODE $= 1$, DIAG need not be set.

Constraint: if MODE $= 2$, DIAG$(i) > 0.0$, for $i = 1, 2, \ldots, n$.

On exit: the scale factors actually used (computed internally if MODE $= 1$).

10: MODE – INTEGER *Input*

On entry: indicates whether or not you have provided scaling factors in DIAG.

If MODE $= 2$ the scaling must have been specified in DIAG.

Otherwise, the variables will be scaled internally.

11: FACTOR – REAL (KIND=nag_wp) *Input*

On entry: a quantity to be used in determining the initial step bound. In most cases, FACTOR should lie between 0.1 and 100.0. (The step bound is FACTOR $\times \|\mathrm{DIAG} \times \mathrm{X}\|_2$ if this is nonzero; otherwise the bound is FACTOR.)

Suggested value: FACTOR $= 100.0$.

Constraint: FACTOR > 0.0.

12: NPRINT – INTEGER *Input*

On entry: indicates whether (and how often) special calls to FCN, with IFLAG set to 0, are to be made for printing purposes.

NPRINT ≤ 0
No calls are made.

NPRINT > 0
FCN is called at the beginning of the first iteration, every NPRINT iterations thereafter and immediately before the return from C05PCF/C05PCA.

13: NFEV – INTEGER *Output*

On exit: the number of calls made to FCN to evaluate the functions.

14: NJEV – INTEGER *Output*

On exit: the number of calls made to FCN to evaluate the Jacobian.

15: R(N $\times$ (N + 1)/2) – REAL (KIND=nag_wp) array *Output*

On exit: the upper triangular matrix R produced by the QR factorization of the final approximate Jacobian, stored row-wise.

16: LR – INTEGER *Dummy*

This parameter is no longer accessed by C05PCF/C05PCA.

17: QTF(N) – REAL (KIND=nag_wp) array *Output*

On exit: the vector $Q^{\mathrm{T}}f$.

18: W(1,1) – REAL (KIND=nag_wp) array *Input*

This parameter is no longer accessed by C05PCF/C05PCA. Workspace is provided internally by dynamic allocation instead.

19: IFAIL – INTEGER *Input/Output*

Note: *for C05PCA, IFAIL does not occur in this position in the parameter list. See the additional parameters described below.*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

Note: *the following are additional parameters for specific use with C05PCA. Users of C05PCF therefore need not read the remainder of this description.*

19: IUSER(*) – INTEGER array *User Workspace*
20: RUSER(*) – REAL (KIND=nag_wp) array *User Workspace*

IUSER and RUSER are not used by C05PCF/C05PCA, but are passed directly to FCN and may be used to pass information to this routine as an alternative to using COMMON global variables.

21: IFAIL – INTEGER *Input/Output*

Note: see the parameter description for IFAIL above.

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL < 0

This indicates an exit from C05PCF/C05PCA because you have set IFLAG negative in FCN. The value of IFAIL will be the same as your setting of IFLAG.

IFAIL $= 1$

On entry, N ≤ 0,
or XTOL < 0.0,
or MAXFEV ≤ 0,
or FACTOR ≤ 0.0,
or LDFJAC $<$ N,
or MODE $= 2$ and DIAG(i) ≤ 0.0 for some i, $i = 1, 2, \ldots,$ N.

IFAIL $= 2$

There have been MAXFEV evaluations of FCN to evaluate the functions. Consider restarting the calculation from the final point held in X.

IFAIL $= 3$

No further improvement in the approximate solution X is possible; XTOL is too small.

IFAIL $= 4$

The iteration is not making good progress, as measured by the improvement from the last five Jacobian evaluations.

IFAIL $= 5$

The iteration is not making good progress, as measured by the improvement from the last ten iterations.

IFAIL $= -999$

Internal memory allocation failed.

The values IFAIL $= 4$ and 5 may indicate that the system does not have a zero, or that the solution is very close to the origin (see Section 7). Otherwise, rerunning C05PCF/C05PCA from a different starting point may avoid the region of difficulty.

7 Accuracy

If $\hat{x}$ is the true solution and D denotes the diagonal matrix whose entries are defined by the array DIAG, then C05PCF/C05PCA tries to ensure that

$$\|D(x - \hat{x})\|_2 \leq \text{XTOL} \times \|D\hat{x}\|_2.$$

If this condition is satisfied with XTOL $= 10^{-k}$, then the larger components of Dx have k significant decimal digits. There is a danger that the smaller components of Dx may have large relative errors, but the fast rate of convergence of C05PCF/C05PCA usually obviates this possibility.

If XTOL is less than ***machine precision*** and the above test is satisfied with the ***machine precision*** in place of XTOL, then the routine exits with IFAIL $= 3$.

Note: this convergence test is based purely on relative error, and may not indicate convergence if the solution is very close to the origin.

The test assumes that the functions and the Jacobian are coded consistently and that the functions are reasonably well behaved. If these conditions are not satisfied, then C05PCF/C05PCA may incorrectly indicate convergence. The coding of the Jacobian can be checked using C05ZAF. If the Jacobian is coded correctly, then the validity of the answer can be checked by rerunning C05PCF/C05PCA with a lower value for XTOL.

8 Further Comments

Local workspace arrays of fixed lengths are allocated internally by C05PCF/C05PCA. The total size of these arrays amounts to $4 \times$ N real elements.

The time required by C05PCF/C05PCA to solve a given problem depends on n, the behaviour of the functions, the accuracy requested and the starting point. The number of arithmetic operations executed by C05PCF/C05PCA is about $11.5 \times n^2$ to process each evaluation of the functions and about $1.3 \times n^3$ to process each evaluation of the Jacobian. Unless FCN can be evaluated quickly, the timing of C05PCF/C05PCA will be strongly influenced by the time spent in FCN.

Ideally the problem should be scaled so that, at the solution, the function values are of comparable magnitude.

9 Example

This example determines the values $x_1, \ldots, x_9$ which satisfy the tridiagonal equations:

$$\begin{aligned} (3-2x_1)x_1 - 2x_2 &= -1, \\ -x_{i-1} + (3-2x_i)x_i - 2x_{i+1} &= -1, \qquad i = 2, 3, \ldots, 8 \\ -x_8 + (3-2x_9)x_9 &= -1. \end{aligned}$$

9.1 Program Text

```
!   C05PCF Example Program Text
!   Mark 24 Release. NAG Copyright 2012.

    Module c05pcfe_mod

!     C05PCF Example Program Module:
!            Parameters and User-defined Routines

!     .. Use Statements ..
      Use nag_library, Only: nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Real (Kind=nag_wp), Parameter        :: factor = 100.0_nag_wp
      Real (Kind=nag_wp), Parameter        :: four = 4.0_nag_wp
      Real (Kind=nag_wp), Parameter        :: one = 1.0_nag_wp
      Real (Kind=nag_wp), Parameter        :: three = 3.0_nag_wp
      Real (Kind=nag_wp), Parameter        :: two = 2.0_nag_wp
      Integer, Parameter                   :: maxfev = 1000, mode = 2, n = 9,  &
                                              nout = 6, nprint = 0
      Integer, Parameter                   :: nr = n*(n+1)/2
      Integer, Parameter                   :: ldfjac = n
    Contains
      Subroutine fcn(n,x,fvec,fjac,ldfjac,iflag)

!       .. Scalar Arguments ..
        Integer, Intent (Inout)              :: iflag
        Integer, Intent (In)                 :: ldfjac, n
!       .. Array Arguments ..
        Real (Kind=nag_wp), Intent (Inout)   :: fjac(ldfjac,n), fvec(n)
        Real (Kind=nag_wp), Intent (In)      :: x(n)
!       .. Local Scalars ..
        Integer                              :: k
!       .. Executable Statements ..
        If (iflag==0) Then

!         Insert print statements here when NPRINT is positive.

          Continue
        Else

          If (iflag/=2) Then
            fvec(1:n) = (three-two*x(1:n))*x(1:n) + one
            fvec(2:n) = fvec(2:n) - x(1:(n-1))
            fvec(1:(n-1)) = fvec(1:(n-1)) - two*x(2:n)
          Else
            fjac(1:n,1:n) = 0.0_nag_wp

            fjac(1,1) = three - four*x(1)
            fjac(1,2) = -two
            Do k = 2, n - 1
              fjac(k,k) = three - four*x(k)
              fjac(k,k-1) = -one
              fjac(k,k+1) = -two
            End Do
            fjac(n,n-1) = -one
            fjac(n,n) = three - four*x(n)

          End If
```

```
        End If

        Return

      End Subroutine fcn
    End Module c05pcfe_mod
    Program c05pcfe

!     C05PCF Example Main Program

!     .. Use Statements ..
      Use nag_library, Only: c05pcf, dnrm2, nag_wp, x02ajf
      Use c05pcfe_mod, Only: factor, fcn, ldfjac, maxfev, mode, n, nout,       &
                             nprint, nr, one
!     .. Implicit None Statement ..
      Implicit None
!     .. Local Scalars ..
      Real (Kind=nag_wp)                  :: fnorm, xtol
      Integer                             :: ifail, j, lr, nfev, njev
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable     :: diag(:), fjac(:,:), fvec(:),     &
                                             qtf(:), r(:), x(:)
      Real (Kind=nag_wp)                  :: w(1,1)
!     .. Intrinsic Procedures ..
      Intrinsic                           :: sqrt
!     .. Executable Statements ..
      Write (nout,*) 'C05PCF Example Program Results'

      Allocate (diag(n),fjac(ldfjac,n),fvec(n),qtf(n),r(nr),x(n))

!     The following starting values provide a rough solution.

      x(1:n) = -one

      xtol = sqrt(x02ajf())
      diag(1:n) = one

      ifail = -1
      Call c05pcf(fcn,n,x,fvec,fjac,ldfjac,xtol,maxfev,diag,mode,factor, &
        nprint,nfev,njev,r,lr,qtf,w,ifail)

      Select Case (ifail)
      Case (0)
!       The NAG name equivalent of dnrm2 is f06ejf
        fnorm = dnrm2(n,fvec,1)
        Write (nout,*)
        Write (nout,99999) 'Final 2-norm of the residuals =', fnorm
        Write (nout,*)
        Write (nout,99998) 'Number of function evaluations =', nfev
        Write (nout,*)
        Write (nout,99998) 'Number of Jacobian evaluations =', njev
        Write (nout,*)
        Write (nout,*) 'Final approximate solution'
        Write (nout,*)
        Write (nout,99997)(x(j),j=1,n)
      Case (2:)
        Write (nout,*)
        Write (nout,*) 'Approximate solution:'
        Write (nout,*)
        Write (nout,99997)(x(j),j=1,n)
      End Select

99999 Format (1X,A,E12.4)
99998 Format (1X,A,I10)
99997 Format (1X,3F12.4)
    End Program c05pcfe
```

9.2 Program Data

None.

9.3 Program Results

```
 C05PCF Example Program Results

 Final 2-norm of the residuals =  0.1193E-07

 Number of function evaluations =        11

 Number of Jacobian evaluations =         1

 Final approximate solution

      -0.5707     -0.6816     -0.7017
      -0.7042     -0.7014     -0.6919
      -0.6658     -0.5960     -0.4164
```